UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL

INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch
TERCEIRA ÁREA

TRANSFORMADA DE FOURIER UTILIZANDO O MAPLE
Décima aula 21/11/2008 (continuação) e Décima primeira aula 24/11/2008

A Transformada de Fourier e a Transformada Inversa

Seja f uma função seccionalmente contínua em cada intervalo finito, definida sobre todo eixo real e tal que $\int_{-\infty}^{\infty} |f(t)|\, dt < \infty$. A transformada de Fourier da função é definida como

F {f(t)}= $\int_{-\infty}^{\infty} f(t)\, e^{-i w t}\, dt$ =F(w) e a transformada de Fourier inversa, a que recupera f(t), é ***F*** ^-1 {F(w)}= $\frac{1}{2\pi}\left(\int_{-\infty}^{\infty} F(w)\, e^{i w t}\, dw\right)$ =f(t)

Exemplo: $f(t) = e^{-|t|}$ ou $f(t) = \begin{cases} e^{-t} & 0 \le t \\ e^{t} & t < 0 \end{cases}$

Vamos resolver este exemplo usando dois recursos do Maple: Primeiro: calculando a integral; Segundo: usando os comandos diretos para a transformada de Fourier e sua inversa.

```
> restart:with(plots):with(inttrans):
```

```
> f(t):=exp(-abs(t));
```

Ou

```
> f:=piecewise(t>=0,exp(-t),t<0,exp(t));
```

```
> plot(f(t),t=-3..3,0..1.5,thickness=2,titlefont=[COURIER,DEFAULT,
  16],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  ytickmarks=[0,1,1.5],axesfont=[COURIER,DEFAULT,16],title=`f(t)=
  exp(-|t|)`);
```

Teste de convergência da integral

```
> teste_conv:=int(abs(f),t = -infinity .. infinity);
```

Note os comandos para a integral, a seguir, Int para indicar a integral e int ou value(Int....) para o cálculo desta integral

```
> F(w):=Int(exp(-abs(t))*exp(-I*w*t),t =-infinity .. infinity);
```

```
> transformada:=value(F(w));
```

```
> F(w):=combine(transformada,trig);
```

A transformada inversa **F** ^-1{F(w)}= $f(t) = \frac{1\left(\int_{-\infty}^{\infty} F(w)\, e^{i w t}\, dw\right)}{2\pi}$

O comando para a integral "Int" foi usado a seguir para que o aluno possa identificar a integral e "expand" para simplificar o integrando

```
> f:= (1/(2*Pi)*Int(transformada*exp(I*w*t),w = -infinity..
  infinity));  f:=expand(f);
```

Agora vamos achar a transformada inversa com o cálculo da última integral, acima, separando os intervalos no tempo t>0 e w>0 e depois t<0 e w<0. Se isto não for feito, o Maple retornará a função integral Ei que é a integral complexa. Para saber mais sobre esta integral, digite >?Ei na linha de comando

```
> assume(t>0);assume(w>0);recuperaf:=(1/(2*Pi)*int(transformada*exp
  (I*w*t),w =-infinity..infinity),"se t>0");
```

```
> assume(t<0);assume(w<0);recuperaf:=(1/(2*Pi)*int(transformada*exp
  (I*w*t),w =-infinity..infinity),"se t<0");
```

O Par Fourier á dado por $F(w) := \int_{-\infty}^{\infty} e^{-|t|}\, e^{-I w t}\, dt$ e

$$f(t) := \frac{1\left(\int_{-\infty}^{\infty} \frac{1\, e^{i w t}}{1+w^2}\, dw\right)}{\pi}$$

Segundo: vamos calcular as transformadas Fourier com os comandos do Maple

```
> restart:with(inttrans):
> f(t):=exp(-abs(t));
> F(w):=fourier(f(t),t,w);
> f:=invfourier(%,w,t);
> convert(f,piecewise);
> modulo_F:=abs(F(w));
> plot(modulo_F,w=-10..10,thickness=2,titlefont=[COURIER,DEFAULT,
  16],labels=["w", "|F(w|"],labelfont=[COURIER,DEFAULT,16],
  ytickmarks=[0,1,1.5],axesfont=[COURIER,DEFAULT,16],title=`|F(w)
  |versus W: O Espectro de Amplitude`);
```

===

Exemplo 2 $f(t) := \begin{cases} k & -a - t < 0 \text{ and } t - a < 0 \\ 0 & otherwise \end{cases}$

```
> restart:with(inttrans):assume(k>0);assume(a>0);interface
  (showassumed=0);
> f(t):=piecewise(t>-a and t<a,k);f:=convert(f(t),Heaviside);
```

Teste de convergência da integral

```
> teste_conv:=int(abs(f),t = -infinity .. infinity);
> F(w):=fourier(f,t,w);
> f:=invfourier(%,w,t);
```

Para os gráficos, vamos atribuir valores para k=2 e a=1

```
> F(w):=subs((a=1,k=2),F(w));
> modulo_F(w):=abs(F(w));
```

O valor máximo do espectro é

```
> limit(modulo_F(w),w=0);
```

Cálculo dos nós: tem-se **aw=nπ, portanto nós em w=nπ/a.** Vamos conferir com a=1, k=2 e n=1

```
> a:=1;limit(modulo_F(w),w=Pi/a);limit(modulo_F(w),w=-Pi/a);
> for i from -3 to  3 do no=i*Pi/a;od;
```

Ou, com o cálculo das raízes de |F(w)| para w=-10..10, tem-se

```
> with(Student[Calculus1]):
  r:=Roots(modulo_F(w),-10..10);pontos:=evalf(r);
```

Para testar, vamos calcular estes limites para |F(w)| com k=2,tem-se a=1 e

```
>  nos_wpos:=seq(limit(modulo_F(w),w=n*Pi/a),n=-3..3);
> plot(modulo_F(w),w=-10..10,thickness=2,titlefont=[COURIER,
  DEFAULT,16],labels=["w", "|F(w)|"],labelfont=[COURIER,DEFAULT,
  16],ytickmarks=[0,1,2,3,4],xtickmarks=[-9.3,-6.3,-3.1,0,3.1,6.2,
  9.3],axesfont=[COURIER,DEFAULT,16],title=`|F(w)|versus W: O
  Espectro de Amplitude`);
```

===

Exemplo 3 $f(t) := \begin{cases} 0 & t < 0 \\ e^{-at} & 0 \leq t \end{cases}$

```
> restart:with(inttrans):assume(a>0);interface(showassumed=0);
> f(t):=piecewise(t<0,0,t>=0,exp(-a*t));f:=convert(f(t),Heaviside);
```

Teste de convergência da integral

```
> teste_conv:=int(abs(f),t = -infinity .. infinity);
> F(w):=fourier(f,t,w);
```

Vamos separar a transformada F(w) em suas parte real e parte imaginária. Isto nos será útil mais adiante, no cáculo do espectro de fase.

```
> A(w):=evalc(Re(F(w)));B(w):=-evalc(Im(F(w)));
> F1(w):=A(w)-B(w)*I;
```

A transformada inversa, a que recupera a função é

```
> f:=invfourier(F(w),w,t);f:=convert(f,piecewise);
```

Para os gráficos, vamos atribuir um valor para a=3

```
> f3:=subs(a=3,f);
> plot(f3,t=-3..3,0..1.5,thickness=3,titlefont=[COURIER,DEFAULT,
  16],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],
  ytickmarks=[0,1,1.5],axesfont=[COURIER,DEFAULT,16],title=`f(t):=
  piecewise(t<0,0,t>=0,exp(-3*t))`);
> F(w):=subs(a=3,F(w));
> modulo_F(w):=abs(F(w));
```

O valor máximo do espectro é

```
> limit(modulo_F(w),w=0);
> plot(modulo_F(w),w=-5..5,thickness=2,titlefont=[COURIER,DEFAULT,
  16],labels=["w", "|F(w)|"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=`|F(w)|versus W: O Espectro
  de Amplitude`);

> Phi:=arctan(-B(w)/A(w));Phi:=evalc(subs(a=3,%));
```

A forma polar de F(w)=$|F(w)|$ e$^{I\Phi}$

Vamos traçar o gráfico da fase e as retas y=Pi/2 e y=-Pi/2 . Para ver o gráfico em escala, clique sobre ele e na régua de status sobre 1:1

```
> plot({Pi/2,-Pi/2,Phi},w=-10..10,thickness=2,titlefont=[COURIER,
  DEFAULT,16],labels=["w", "Phi"],labelfont=[COURIER,DEFAULT,16],
  axesfont=[COURIER,DEFAULT,16],title=` O Espectro de Fase`);
```

==FIM============================== ==================

A seguir: as Transformadas Fourier Seno e Co-seno

As Transformadas Fourier Seno e Co-seno e comparação com as integrais seno e co-seno

(Estas transformadas têm origem na representação integral seno e co-seno (veja o arquivo integral Fourier Trigonométrica) e, como elas, em analogia às séries Fourier seno e co-seno para funções periódicas definidas em meio intervalo)

Se f é uma função seccionalmente contínua em cada intervalo finito, **definida sobre o eixo real positivo** e tal que $\int_0^\infty |f(t)|\,dt < \infty$, então:

as transformadas seno e co-seno da função f são definidas, respectivamente, por:

F s{f(t)}= $\int_{0}^{\infty} f(t)\sin(w\,t)\,dt$ =Fs(w) e ***F*** c{f(t)}= $\int_{0}^{\infty} f(t)\cos(w\,t)\,dt$ =Fc (w). **(1)**

As **transformadas inversas**, aquelas que recuperam a função no domínio tempo, são

F s^-1{f(t)} $= f(t) = \dfrac{2\left(\int_{0}^{\infty} F_s(w)\sin(w\,t)\,dw\right)}{\pi}$ e ***F*** c^-1{f(t)}= $f(t) = \dfrac{2\left(\int_{0}^{\infty} F_c(w)\cos(w\,t)\,dw\right)}{\pi}$, respectivamente.

Lembrar:

A **integral Fourier Trigonométrica:** a reperesentação integral Fourier seno da função estendida como ímpar no intervalo [-L,0) e arepresentação integral co-seno da função estendida como par no intervalo [-L,0) são definidas como $\frac{1}{\pi}\int_{0}^{\infty} B(w)\sin(w\,t)\,dw$ e $\frac{1}{\pi}\int_{0}^{\infty} A(w)\cos(w\,t)\,dw$

onde $A(w) = \int_{-\infty}^{\infty} f(t)\cos(w\,t)\,dt = 2\left(\int_{0}^{\infty} f(t)\cos(w\,t)\,dt\right)$ e $B(w) = \int_{-\infty}^{\infty} f(t)\sin(w\,t)\,dt = 2\left(\int_{0}^{\infty} f(t)\sin(w\,t)\,dt\right)$ **(2)**

No caso das transformadas, pode-se inferir, comparando (1) e (2) que **B(w)=2Fs(w) e A(w)=2Fc (w), ou seja:**

a representação integral Fourier seno da função estendida como ímpar em [-L,0) pode ser escrita como f(t)= $\frac{2}{\pi}\int_{0}^{\infty} F_s(w)\sin(w\,t)\,dw$=***F*** ^-1{Fs(w)}

e a representação integral co-seno da função estendida como par em [-L,0) como f(t) = $\frac{2}{\pi}\int_{0}^{\infty} F_c(w)\cos(w\,t)\,dw$=***F*** ^-1{Fc(w)}

EXEMPLO: Seja $f(t) := \begin{cases} 1 & -t \le 0 \text{ and } t - k < 0 \\ 0 & \textit{otherwise} \end{cases}$. Esta função nào é par, nem é ímpar no intervalo t>0, onde está definida.

```
> restart:with(inttrans):assume(t>=0);assume(k>0);interface
  (showassumed=0);
> f(t):=piecewise(t>=0 and t<k,1,t>k,0); f1:=convert(f(t),
  Heaviside);
```

Para o gráfico, vamos fazer k=4

```
> f4:=subs(k=4,f(t));
> plot(f4,t=0..6,thickness=3,titlefont=[COURIER,DEFAULT,10],labels=
  ["t", "f(t)"],labelfont=[COURIER,DEFAULT,16],ytickmarks=[0,1,
  1.5],axesfont=[COURIER,DEFAULT,16],title=`f(t)=1 se t>=0 e t<4, f
  (t)=0 p/outros valores de t`);
```

Teste de convergência da integral

```
> teste_conv:=int(abs(f(t)),t =0 .. infinity);
> Fs(w):=simplify(int(f1*sin(w*t),t = 0 .. infinity));
> Fc(w):=simplify(int(f1*cos(w*t),t = 0 .. infinity));
```

Vamos escrever a transformada na forma integral como F(w) = $\frac{1\,[A(w) - B(w)\,I]}{2}$ =Fc(w)-Fs(w)*I , a soma de sua parte real (correspondente a transformada co-seno Re{F(w)}) com a parte imaginária (correspondente a transformada seno Im{F(w)}) .

```
> F(w):=Fc(w)-Fs(w)*I;
> F_inv:=invfourier(F(w),w,t);
> f:=convert(F_inv,piecewise,t);
> modulo_F(w):=sqrt(Fc(w)^2+Fs(w)^2);
```

O valor máximo do espectro é

```
> limit(modulo_F(w),w=0);
```

Como antes, para o gráfico, vamos fazer k=4

```
> modulo_F4:=subs(k=4,modulo_F(w));
```

Cálculo dos nós: portanto nós nas raízes de |F(w)| Vamos conferir com k=4

```
> with(Student[Calculus1]):
  r:=Roots(modulo_F4,-5..5);Digits:=4:pontos:=evalf(r);
```

ou, utilizando a fórmula, nós em aw=nπ, sendo a=2 (metade da largura da fenda)

```
> for i from -3 to  3 do no=i*Pi/2;od;
```

Para testar, vamos calcular estes limites para |F(w)|

```
>  nos_wpos:=seq(limit(modulo_F4,w=n*Pi/2),n=-3..3);
> plot(modulo_F4,w=-5..5,thickness=2,titlefont=[COURIER,DEFAULT,
  16],labels=["w", "|F(w)|"],labelfont=[COURIER,DEFAULT,16],
  xtickmarks= [-4.713, -3.142, -1.571, 1.571, 3.142, 4.713],
  axesfont=[COURIER,DEFAULT,8],title=`|F(w)|versus W: O Espectro de
  Amplitude`);
```

=========================

Agora, vamos calcular a integral Fourier seno, ou seja, B(w) e a integral co-seno a(w) para comparar os resultados obtidos e confirmar o que inferimos acima. Entretanto, deve-se multiplicar por $\sqrt{2}\,\sqrt{\pi}$ para obter as f'órmulas convencionadas na disciplina.

```
> restart:with(inttrans):assume(t>=0);assume(k>0);interface
  (showassumed=0);
> f(t):=piecewise(t>=0 and t<k,1,t>k,0); f1:=convert(f(t),
  Heaviside);
> A(w):=2*int(f(t)*cos(w*t),t=0 .. infinity);invfourier(A(w),w,t);
> B(w):=2*int(f(t)*sin(w*t),t = 0 .. infinity);invfourier(A(w),w,t)
  ;
```

Tinhamos obtido nas transformadas $Fs(w) := -\frac{-1+\cos(w\,k)}{w}$ e $Fc(w) := \frac{\sin(w\,k)}{w}$, portanto,estes valores comprovam a afirmação feita acima

A transformada inversa de A(w) recupera a extensão par de f(t) e a tranformada inversa de B(w) recupera a função estendida como ímpar, conforme pode ser constatado a seguir com o cálculo das transformadas inversas e seus respectivos gráficos.

```
> fa:=subs(k=4,invfourier(A(w),w,-t));
> plot(fa,t=-4..4,discont=true,thickness=2,titlefont=[COURIER,
  DEFAULT,16],title=`A transformada inversa de A(w)`);

> fb:=subs(k=4,invfourier(B(w)*I,w,-t));

> plot(fb,t=-4..4,discont=true,thickness=2,titlefont=[COURIER,
  DEFAULT,16], title=`A transformada inversa de B(w)`);
```